



Using ODS Options with SAS Stored Processes


THE POWER TO KNOW.®

Cynthia L. Zender
SAS Institute, Inc.

Stored Process Consumers Don't F3




Stored Process Consumers Don't Do Commands



3

Stored Process Consumers Don't SUB, RUN, REC or CLE




4

Stored Process Consumers Don't Do ODS



5

Stored Process Consumers Use Client Applications



6

Today's Topics

Concrete examples of converting existing SAS programs to stored processes:

- Basic program conversion
- Defining user-supplied input parameters
- Conversion of ODS options
- Conversion of SAS/Graph programs
- Brief discussion of streaming results versus transient and permanent result packages

7

Stored Process Basic Concepts

SAS stored processes (SAS programs) can be executed by different SAS Intelligence Platform application.

These applications are called *client applications*.

A computer that exists to service processing requests from client applications is called a *server*. The SAS Intelligence Platform uses two different servers to execute stored processes: *Stored Process server* or a *Workspace server*.

Stored process programs can accept information that is provided by the stored process consumer at the time of stored process execution. This information takes the form of *input parameters*.

8

Steps To Manual Program Conversion

1. Have a working SAS program.
2. Use a text editor to convert your SAS program to be a stored process by adding the *ProcessBody; comment, %GLOBAL statement for input parameters and %STPBEGIN; and %STPEND; macro invocations in place of the ODS invocation statements.
3. Register the stored process metadata relevant to the stored process program using SAS Management Console.
4. Test your stored process in all client applications from which it will be executed.

9

Program Conversion Using SAS Enterprise Guide

1. Have a working SAS program, EG project or EG task.
2. Use the **Stored Process Wizard** to convert and register your new stored process. The Stored Process Wizard puts all the required SAS Macro Facility statements into your program and scans the program for possible input parameters. (You may or may not need to modify the code generated by EG.)
3. Test the stored process in all client applications from which it will be executed.

10

Data

The stored processes use **SASHELP.SHOES**

- Easy to use
- Most sites have installed
- Demonstration programs and stored processes will run with least amount of setup.

11

Step 1: Working SAS Program

Proc Format for traffic lighting:

```
proc format;  
  value incnt 0-5000='light yellow'  
            .='light yellow'  
            50000-high='light green'  
            other='white';  
  
run;
```

12

Step 1: Working SAS Program

Set parameter, ODS Invocation and PROC SORT:

```
%let wantreg=Canada;

options nodate nonumber missing='0'
       orientation=landscape;
ods rtf file='c:\temp\demo1.rtf'
      bodytitle startpage=no keepn notoc_data;
ods pdf file='c:\temp\demo1.pdf'
      bookmarkgen=no compress=9 startpage=no;
ods html file='c:\temp\demo1.html'
      style=sasweb rs=none;
ods escapechar='^';

proc sort data=sashelp.shoes out=shoes;
  by Region; where Region = "&wantreg";
run;
```

13

Step 1: Working SAS Program

Proc Report Step

```
proc report data=shoes nowd;
  title "^S={Foreground=Purple font_size=16pt}&wantreg";
  title2 "Region Report";
  column region product sales;
  define region /group noprint;
  define product / group;
  define sales /sum f=dollar14.;
  break after region / summarize style=Header;
run;
```

14

Step 1: Working SAS Program

Use ODS PDF TEXT=

```
%let ss=^S={just=c font_size=13pt font_weight=Bold
          font_style=italic};
ods pdf text="^2n ^_^S={} &ss";
ods pdf text="&ss.Subsidiary Performance Report";
```

The STARTPAGE=NO option suppresses the PROC TABULATE title in PDF results, so the TEXT= string will be used in the PDF results to simulate a TITLE. The ODS ESCAPECHAR string is formatting this text to be the same size and font as the SAS System Title.

15

Step 1: Working SAS Program

Proc Tabulate Step

```
proc tabulate data=shoes f=dollar14.;
  title 'Subsidiary Performance Report';
  var sales;
  class subsidiary product;
  table subsidiary all*{s={background=white}},
         Sales*{style={background=incent.}}
         *(product=' '
            all='Total'*{S={background=white}})
         / style_precedence=row;
  keylabel sum=' '
            all='Subsidiary Total';
  label Sales='Product';
run;

ods _all_ close;
```

16

Program Results HTML

**Canada
Region Report**

Product	Total Sales
Boot	\$385,613
Men's Casual	\$441,903
Men's Dress	\$920,101
Sandal	\$14,798
Slipper	\$952,751
Sport Shoe	\$140,389
Women's Casual	\$410,807
Women's Dress	\$989,350
	\$4,255,712

Partial results

Subsidiary Performance Report

Subsidiary	Product								Total
	Boot	Men's Casual	Men's Dress	Sandal	Slipper	Sport Shoe	Women's Casual	Women's Dress	
Calgary	\$17,720	0	\$12,775	\$2,886	\$5,676	\$9,745	0	\$12,601	\$61,403
Montreal	\$40,213	\$53,929	\$112,009	\$3,002	\$135,305	\$29,435	\$24,497	\$132,638	\$531,028

17

Program Results RTF

**Canada
Region Report**

Product	Total Sales
Boot	\$385,613
Men's Casual	\$441,903
Men's Dress	\$920,101
Sandal	\$14,798
Slipper	\$952,751
Sport Shoe	\$140,389
Women's Casual	\$410,807
Women's Dress	\$989,350
	\$4,255,712

Subsidiary Performance Report

Subsidiary	Product								Total
	Boot	Men's Casual	Men's Dress	Sandal	Slipper	Sport Shoe	Women's Casual	Women's Dress	
Calgary	\$17,720	0	\$12,775	\$2,886	\$5,676	\$9,745	0	\$12,601	\$61,403
Montreal	\$40,213	\$53,929	\$112,009	\$3,002	\$135,305	\$29,435	\$24,497	\$132,638	\$531,028
Ottawa	\$7,892	\$19,210	0	\$2,600	\$30,905	\$2,398	\$18,712	\$33,824	\$115,241
Toronto	\$33,291	\$15,403	\$37,519	\$1,190	\$80,352	\$34,585	\$63,492	\$33,040	\$319,272
Vancouver	\$286,897	\$353,361	\$757,298	\$5,120	\$700,513	\$64,026	\$304,106	\$756,347	\$3,227,768
Subsidiary Total	\$385,613	\$441,903	\$920,101	\$14,798	\$952,751	\$140,389	\$410,807	\$989,350	\$4,255,712

Effect of BODYTITLE
Suppressed page break

18

Program Results PDF

*Canada
Region Report*

Product	Total Sales
Boot	\$385,613
Men's Casual	\$441,903
Men's Dress	\$920,101
Sandal	\$14,798
Slipper	\$992,751
Sport Shoe	\$140,389
Women's Casual	\$410,807
Women's Dress	\$989,350
	\$4,255,712

Subsidiary Performance Report

Subsidiary	Product								
	Boot	Men's Casual	Men's Dress	Sandal	Slipper	Sport Shoe	Women's Casual	Women's Dress	Total
Calgary	\$17,720	0	\$12,775	\$2,886	\$5,676	\$9,745	0	\$12,601	\$61,403
Montreal	\$40,213	\$53,929	\$112,009	\$3,002	\$135,305	\$29,433	\$24,497	\$132,638	\$531,028
Ottawa	\$7,892	\$19,210	0	\$2,600	\$30,905	\$2,598	\$16,712	\$33,824	\$115,741
Toronto	\$33,291	\$15,403	\$37,519	\$1,190	\$80,352	\$34,583	\$63,492	\$53,940	\$319,772
Vancouver	\$286,497	\$353,361	\$757,798	\$5,120	\$700,313	\$64,026	\$304,106	\$756,341	\$3,237,768
Subsidiary Total	\$385,613	\$441,903	\$920,101	\$14,798	\$992,751	\$140,389	\$410,807	\$989,350	\$4,255,712

Suppressed page break
ODS PDF TEXT= string

19

Step 2: Convert SAS Program

- Take out %LET statement for any input parameters that will be supplied by the stored process consumer working in the client application.
- Add %GLOBAL statement to ensure that input parameters are GLOBAL macro variables in the Global Macro Symbol Table.
- Add *ProcessBody; comment.
- Remove ODS "sandwich" statements.
- Replace with %STPBEGIN/%STPEND macro calls.
- Add any ODS option overrides **before** %STPBEGIN macro call.

20

Stored Process Parameters and ODS

ODS Options and Stored Process Reserved Variables	
ODS Task or Option	Equivalent Reserved Parameter Name
Destination	_ODSDEST
FILE=	generally not used with %STPBEGIN/%STPEND for most clients because _WEBOUT is the reserved fileref for stored process output.
STYLE=	_ODSSTYLE sets the STYLE= option to an ODS style valid on the server running the stored process
STYLESHEET=(URL=)	_ODSSTYLESHEET sets the ODS STYLESHEET option (HTML-based output)
Specify other options to be appended to the ODS statement	_ODSOPTIONS

21

Step 2: Convert SAS Program

```
%let wantreg=Canada;

options nodate nonumber missing='0'
      orientation=landscape;
ods rtf file='c:\temp\demo1.rtf'
      bodytitle startpage=no keepn notoc_data;
ods pdf file='c:\temp\demo1.pdf'
      bookmarkgen=no compress=9 startpage=no;
ods html file='c:\temp\demo1.html'
      style=sasweb rs=none;
ods escapechar='^';
. . . SAS code . . .
ods _all_ close;
```

Your stored process is not creating FILES, it is creating a single result stream or temporary result file. The person who executes the stored process can save the results to a local file, if required. Your best bet to convert this program is to start by creating 3 stored processes.

22

Step 2: Convert SAS Program

```
%let wantreg=Canada;
options nodate nonumber missing='0'
      orientation=landscape;
ods rtf file='c:\temp\demo1.rtf'
      bodytitle startpage=no keepn notoc_data;
ods pdf file='c:\temp\demo1.pdf'
      bookmarkgen=no compress=9 startpage=no;
ods html file='c:\temp\demo1.html'
      style=sasweb rs=none;
ods escapechar='^';

. . . SAS code . . .

ods _all_ close;
```

***ProcessBody;**
%global . . .;

%LET . . .;

%STPBEGIN;

%STPEND;

%STPBEGIN/%STPEND will replace your ODS "sandwich".

23

ODS Reserved Macro Variables

- The ODS reserved macro variables that you can use as input parameters are defined as global macro variables.
- Must be specified before the %STPBEGIN macro if you want to override the default values.
- The _ODSDEST, _ODSSTYLE, and _ODSSTYLESHEET macro variables are **usually** set for your stored process based on which client application has requested execution of the stored process.

24

_ODSDEST and HTML Results


- Most SAS Intelligence Platform applications can receive HTML results. The _ODSDEST reserved macro variable is set to HTML by default for SAS Enterprise Guide and the SAS Information Delivery Portal.



25

_ODSDEST and SAS Report XML Results

In general, all SAS Intelligence Platform applications can receive SAS Report XML results. SAS Report XML is the default _ODSDEST value for the SAS Add-In for Microsoft Office and SAS Web Report Studio.



26

...

Why Override ODS Option Values?

- Stored process author wants to control how results are returned to the end-user.
- Stored process program is only meant to run in certain SAS Intelligence Platform client applications.
- Stored process author does NOT want end-user to use client application settings to determine how results are returned to the end-user.

27

Step 2: Create HTML Stored Process

Original ODS HTML Statements	Modified Code
<pre>ods html file='c:\temp\demo1.html' style=sasweb rs=none; . . . SAS code . . . ods _all_ close;</pre>	<pre>%global wantreg _odsoptions _odsstyle _odsstylesheet; *ProcessBody; %let _odsoptions = rs=none; %let _odsstyle = sasweb; %let _odsstylesheet=; %stpbegin; . . . SAS Code . . . %stpend;</pre>

DEMO2_HTML.SAS

28

Step 2: Create RTF Stored Process

Original ODS RTF Statements	Modified Code
<pre>ods rtf file='c:\temp\demo1.rtf' bodytitle startpage=no keepn notoc_data; . . . SAS code . . . ods _all_ close;</pre>	<pre>%global wantreg _odsoptions _odsdest _odsstyle _odsstylesheet; *ProcessBody; %let _odsdest=rtf; %let _odsstyle=rtf; %let _odsstylesheet=; %let _odsoptions = bodytitle startpage=no keepn notoc_data; %stpbegin; . . . SAS Code . . . %stpend;</pre>

DEMO3_RTF.SAS

29

Step 2: Create PDF Stored Process

Original ODS HTML Statements	Modified Code
<pre>ods pdf file='c:\temp\demo1.pdf' bodytitle startpage=no keepn notoc_data; . . . PROC REPORT . . . ODS PDF TEXT= . . .; . . . PROC TABULATE . . . ods _all_ close;</pre>	<pre>%global wantreg _odsoptions _odsdest _odsstyle _odsstylesheet; *ProcessBody; %let _odsdest=pdf; %let _odsstyle=printer; %let _odsstylesheet=; %let _odsoptions = bookmarkgen=no compress=9 startpage=no; %stpbegin; . . . PROC REPORT . . . ODS PDF TEXT= . . .; . . . PROC TABULATE . . . %stpend;</pre>

DEMO4_PDF.SAS

30

Step 3: Register the Metadata

- Stored Process Name

demo2_html Properties

General | Execution | Parameters | Authorization

Name: demo2_html

Type: Stored Process

Description:

Folder: STP_Orion

Created: 1/20/07 5:24 PM

Modified: 1/20/07 5:24 PM

Keywords:

Responsibilities:

Name	Role
------	------

OK Cancel Help

31

Step 3: Register the Metadata

- Execution Environment

demo2_html Properties

General | Execution | Parameters | Authorization

SAS server: SASMain - Logical Stored Process Server

Source code repository: S:\Workshop\winsas\sbisp

Source file: demo2_html.sas

Input: None


Output: Streaming

OK Cancel Help

32

Step 3: Register the Metadata


- Parameters



33

Step 3: Register the Metadata


Register the parameters



34

Step 3: Register the Metadata


All the stored processes registered:



35

Step 4: Test the Stored Processes

Use SAS Enterprise Guide to test.



36

Testing In Other Client Applications

Client Application	Demo2_HTML	Demo3_RTF	Demo4_PDF
SAS Enterprise Guide	Y	Y	Y
SAS Information Delivery Portal	Y	Y	Y
SAS Add-in Word	Y	Y	N/A
SAS Add-in Excel	Y	N/A	N/A
SAS Add-in PowerPoint	Y* (SRX)	N/A	N/A
SAS Web Report Studio	Y* (SRX)	N/A	N/A

DEMO2_HTML Stored process runs in PowerPoint and Web Report Studio, only if the override for _ODSOPTIONS is removed because RS=NONE is not a valid option for SAS Report Model XML. Without the override, the %STPBEGIN/%STPEND returns SAS Report Model XML to these two client applications.

BONUS: DEMO5_MACRO.SAS

Register this program as two different stored processes:


- Demo5_Macro has just an input parameter for Region. Use this stored process if you want to allow the end-user to pick the result type using the client application options.
- Demo6_Macro has input parameters for Region and Destination. Use this stored process if you want to allow the end-use to pick the result type using the property sheet interface.

The program contains Macro Facility logic that tests the destination the user supplies and sets options and output appropriately.

38

Executing Demo5_Macro Stored Process


First, change client application to return RTF output:



39

Executing Demo5_Macro Stored Process


Next, execute the stored process and select Eastern Europe for Region, when prompted:



40

Executing Demo5_Macro Stored Process

Finally, review the RTF results:



The screenshot shows the SAS Project Designer interface. On the left, the Project Explorer shows a tree view with 'RTF : demo5_macro' selected. The main window displays the RTF output, which includes a title 'Eastern Europe Region Report' and a table of product sales.

Product	Total Sales
Boot	\$306,785
Men's Casual	\$576,396
Men's Dress	\$325,761
Sandal	\$3,716
Slipper	\$509,698
Sport Shoe	\$91,202
Women's Casual	\$209,256
Women's Dress	\$362,126
	\$2,394,940

41

How The Stored Process Works

```
*ProcessBody;  
%setopt;  
%stpbegin;  
%RegRept;  
%stpend;
```

42

How The Stored Process Works

***ProcessBody;**

%setopt;

%stpbegin;

%RegRept;

%stpend;

```
demo5_macro.sas - Notepad
File Edit Format View Help

%global wantreg _odsoptions _odsdest
_odsstyle _odsstylesheet;

%macro setopt;
%if %upcase(&_odsdest) = RTF %then %do;
  %let _odsstyle=rtf;
  %let _odsstylesheet=;
  %let _odsoptions=bodytitle startpage=no nokeepn notoc_data;
%end;
%else %if %upcase(&_odsdest) = PDF %then %do;
  %let _odsstyle=printer;
  %let _odsstylesheet=;
  %let _odsoptions=bookmarkgen=no compress=9 startpage=no;
%end;
%else %if %upcase(&_odsdest) = HTML %then %do;
  %let _odsstyle=sasweb;
  %let _odsstylesheet=;
  %let _odsoptions= rs=none;
%end;
%else %if %upcase(&_odsdest)=TAGSETS.SASREPORT11 %then %do;
  %let _odsstyle=normal;
  %let _odsstylesheet=;
  %let _odsoptions=;
%end;

options nodate nonumber missing='0' orientation=landscape;
%mend setopt;
```

43

How The Stored Process Works

***ProcessBody;**

%setopt;

%stpbegin;

%RegRept;

%stpend;

```
%macro RegRept;
ods escapechar='^';

proc sort data=sashelp.shoes out=shoes;
  by region;
  where region = "&wantreg";
run;

proc format;
  value incnt 0-5000='light yellow'
  .='light yellow'
  5000-high='light green'
  other='white';
run;

proc report data=shoes nowd;
  title "AS={Foreground=Purple font_size=16pt}&wantreg";
  title2 "region Report";
  column region product sales;
  define region /group noprint;
  define product / group;
  define sales /sum f=dollar14.;
  break after region / summarize style=Header;
run;

%if %upcase(&_odsdest)=PDF %then %do;
  %let ss=AS={just=c font_size=13pt font_weight=Bold font_style=italic};
  ods pdf text="A2n A_AS={ } &ss";
  ods pdf text="&ss.Subsidiary Performance Report";
%end;

proc tabulate data=shoes f=dollar14.;
  title "Subsidiary Performance Report";
  var sales;
  class subsidiary product;
  table subsidiary all*{s={background=white}},
  sales*{style={background=incnt,}}
  *(product= all=Total*{s={background=white}})
  / style_precedence=row ;
  keylabel sum=
  all="&wantreg Total";
  label sales="Product";
run;

%mend RegRept;
```

44

Convert A SAS/Graph Program

```
ods html path='c:\temp\' (url=none)
      file='demo1_graph.html' style=analysis;
options device=actximg;
. . . Other Report Code . . .
proc gchart data=shoes;
  title "&wantreg: Subsidiary Product Analysis";
  vbar subsidiary /type=sum sumvar=sales
      maxis=axis1 frame woutline=1 raxis=axis2
      legend=legend1 subgroup=product;
run;
quit;
ods html close;
```

45

Convert A SAS/Graph Program

When you convert a SAS/Graph program, you have several decisions to make that are unique to the production of graphical output:

- What device driver should be used to create your output?
- What graphics options need to be supplied in overrides to %STPBEGIN macro call?
- What stored process output type to choose?

46

SAS/GRAPH Reserved Variables

SAS/GRAPH Options and Stored Process Reserved Variables

Graphics Options	Reserved Parameter Name
DEVICE=	_GOPT_DEVICE
HSIZE=	_GOPT_HSIZE
VSIZE=	_GOPT_VSIZE
XPIXELS=	_GOPT_XPIXELS
YPIXELS=	_GOPT_YPIXELS
Other GOPTIONS	_GOPTIONS sets any SAS/GRAPH option valid in the GOPTIONS statement

The ACTXIMG driver used in the demonstration program produces "static" output that looks like the ACTIVEX output, but does not have the interactivity of ACTIVEX output. If you wanted to use a different device driver, you would need to specify an override for _GOPT_DEVICE.

47

Stored Process Execution

- The stored process output type setting depends on the server type.

Stored Process Server


Workspace Server

Note that STREAMING output is not available on the Workspace Server. The Stored Process server is most like the server used to execute Application Dispatcher programs in SAS/IntrNet applications.

48

Stored Process Execution


SAS Enterprise Guide Results



49

Stored Process Execution


SAS Enterprise Guide results versus SAS Add-in for Microsoft Office results



50

Stored Process Execution


In the execution properties for the stored process, we chose STREAMING output:



51

Stored Process Execution


SAS Enterprise Guide forces the STREAMING output type to be TRANSIENT output type. But the SAS Add-in for Microsoft Office does not have this option:



52

Stored Process Execution


Set result type to transient



53

Stored Process Execution

Test in Microsoft Word:



54

Understanding Transient Packages

- The transient package result type is returned to a temporary directory location on the client machine (similar to temporary Internet cache file used with a browser).

Transient package output returns a temporary package to the client.

55

Understanding Permanent Packages

- The permanent package is also returned to the stored process client application, but it can be written to a webDAV repository or to the server file system.

Setting up a webDAV repository is an advanced topic and falls outside the scope of this paper.

56

Downloading the SAS Package Reader



The SAS Package Reader may be installed when the SAS Intelligence Platform software is installed. If you are going to create permanent packages, you will need the SAS Package Reader to open the .SPK files.

57

About the Speaker

Speaker Cynthia Zender

Company SAS Institute, Inc.

Telephone (919) 531-9012

Comments & E-Mail Cynthia.Zender@sas.com

This paper and sample programs are available for download at:

<http://support.sas.com/rnd/papers>

58